# Maintainer: Evangelos Foutras <evangelos@foutrelis.com>
# Contributor: Xavier Devlamynck <magicrhesus@ouranos.be>

pkgname=garcon
pkgver=4.16.1
pkgrel=1
pkgdesc="Implementation of the freedesktop.org menu specification"
arch=('x86_64')
url="https://www.xfce.org/"
license=('LGPL')
groups=('xfce4')
depends=('libxfce4ui')
makedepends=('intltool' 'python' 'gobject-introspection')
source=(https://archive.xfce.org/src/xfce/garcon/${pkgver%.*}/$pkgname-$pkgver.tar.bz2)
sha256sums=('84914927a2c1fda167f67af26a6640630a744a22940df508aa6c752cdbd3d21d')

build() {
  cd "$srcdir/$pkgname-$pkgver"

  ./configure \
    --prefix=/usr \
    --sysconfdir=/etc \
    --libexecdir=/usr/lib \
    --localstatedir=/var \
    --disable-static \
    --disable-debug
  make
}

package() {
  cd "$srcdir/$pkgname-$pkgver"
  make DESTDIR="$pkgdir" install
}

# vim:set ts=2 sw=2 et:
